

Soluções Laboratório 1

Questão 1 (Empresa de aço)

O sistema em formato CPLEX lp é

```
maximize 25p+30c
subject to
    7p+10c <= 56000
bounds
    0 <= p <= 6000
    0 <= c <= 4000
end
```

e o GLPK responde com

Objective: obj = 192000 (MAXimum)

No.	Row name	St	Activity	Lower bound	Upper bound	Marginal
1	r.3	NU	56000		56000	3
No.	Column name	St	Activity	Lower bound	Upper bound	Marginal
1	p	NU	6000	0	6000	4
2	c	B	1400	0	4000	

Questão 2 (Questão 6 da lista 2)

Sejam $V = \{V_1, V_2\}$ e $NV = \{NV_1, NV_2, NV_3\}$ os conjuntos de óleos vegetais e não vegetais e $O = V \cup NV$ o conjunto de todos óleos. Seja ainda c_i o custo por tonelada do óleo $i \in O$ e a_i a acidez do óleo $i \in O$. (Por exemplo $c_{V_1} = 110$ e $a_{NV_2} = 4.2$.) Com variáveis x_i (toneladas refinadas do óleo $i \in O$) e x_o (quantidade total de óleo produzido) podemos formular

$$\begin{aligned}
 &\text{maximiza} && 150x_o - \sum_{i \in O} c_i x_i \\
 &\text{sujeito a} && \sum_{i \in V} x_i \leq 200 && \text{limite óleos vegetais} \\
 &&& \sum_{i \in NV} x_i \leq 250 && \text{limite óleos não vegetais} \\
 &&& 3x_o \leq \sum_{i \in O} a_i x_i \leq 6x_o && \text{Intervalo acidez} \\
 &&& \sum_{i \in O} x_i = x_o && \text{Óleo total} \\
 &&& x_o, x_i \geq 0 && \forall i \in O.
 \end{aligned}$$

Uma solução explícita em CPLEX lp usando os dados da questão é

```
max 150xo-110xv1-120xv2-130xnv1-110xnv2-115xnv3
st
xv1+xv2 <= 200
xnv1+xnv2+xnv3 <= 250
3xo-8.8xv1-6.1xv2-2.0xnv1-4.2xnv2-5.0xnv3<=0
8.8xv1+6.1xv2+2.0xnv1+4.2xnv2+5.0xnv3-6xo<=0
xv1+xv2+xnv1+xnv2+xnv3-xo=0
end
```

Uma solução em AMPL é

```
# conjuntos de óleos
set V := { "V1", "V2" };
set NV := { "NV1", "NV2", "NV3" };
set O := V union NV;
# custo e acidez
param c { O };
param a { O };
# quantidades
var x { O } >= 0;
var xo >= 0;

maximize lucro: 150*xo - sum { i in O } c[i]*x[i];
subject to limitveg: sum { i in V } x[i] <= 200;
subject to limitnveg: sum { i in NV } x[i] <= 250;
subject to acidez1: 3*xo <= sum { i in O } a[i]*x[i];
subject to acidez2: sum { i in O } a[i]*x[i] <= 6*xo;
subject to oleatot: sum { i in O } x[i] = xo;

data;

param c := "V1" 110, "V2" 120, "NV1" 130, "NV2" 110, "NV3" 115;
param a := "V1" 8.8, "V2" 6.1, "NV1" 2.0, "NV2" 4.2, "NV3" 5.0;
end;
```

Questão 3

Uma formulação em CPLEX lp é

```
max na+nb+nc
st
xa+xb+xc = 100
2na-0.1xa = 6
2nb-0.2xb = 7
2nc-0.3xc = 10
bounds
5 <= na <= 10
5 <= nb <= 10
5 <= nc <= 10
end
```

outra em MathProg é

```
# disciplinas
set D := { "A", "B", "C" };

# nota prova 1
param prova1 { D };
# nota por hora de estudo na prova 2
param notaporhora { D };

# variáveis para notas
var n { D } >= 0;
# variáveis para horas de estudo
```

```
var x { D } >= 0;

maximize notaMedia: sum { d0 in D } n[d0];

subject to CemHoras: sum { d0 in D } x[d0] = 100;

subject to NotaResultado { d0 in D }:
    n[d0] = (prova1[d0]+notaporhora[d0]*x[d0])/2;

subject to NotaMinima { d0 in D }:
    n[d0] >= 5;

subject to NotaMaxima { d0 in D }:
    n[d0] <= 10;

data;

param notaporhora := "A" 0.1, "B" 0.2, "C" 0.3;
param prova1 := "A" 6, "B" 7, "C" 10;
end;
```

Questão 4 (McDonalds)

O modelo está disponível em

<http://www.inf.ufrgs.br/~mrpritt/oc/McDonalds.mod>

- Uma dieta ótima custa 24.31 R\$ e consiste (aproximadamente) em 5 McDuplos, três maçãs e duas casquinhas mistas.
- A dieta ótima vegetariana custo 39.63 R\$ e consiste (aproximadamente) em 7 molhos salada Ranch, 14 casquinhas baunilha, 2 casquinhas chocolate e 5 maçãs.
- Nesse caso a dieta ótima custa 31.56 R\$ e consiste (aproximadamente) em um quarterão com queijo, um McDuplo, um Cheeseburger, um Hamburger, um McBacon Junior, McFritas Grande, meio McFrita média, uma casquinha mista e um maçã.

Bom apetite!

Questão 5 (Companhia de torneiras)

Com as variáveis definidas um conjunto de fábricas $F = [2]$ e de armazens $A = [2]$ podemos formular

$$\text{minimiza} \quad \sum_{i \in F} p_i(x_i + s_i + y_i) + \sum_{j \in A} w_j h_j + \sum_{i,j} c_{ij}(z_{ij} + v_{ij}) \quad (1)$$

$$\text{sujeito a} \quad x_i + s_i \leq L_i \quad \forall i \in F \quad (2)$$

$$y_i \leq L_i \quad \forall i \in F \quad (3)$$

$$\sum_{i \in F} z_{ij} \geq R_{1j} \quad \forall j \in A \quad (4)$$

$$\sum_{i \in F} z_{ij} + v_{ij} \geq R_{2j} \quad \forall j \in A \quad (5)$$

$$x_1 + s_1 + y_1 \geq 0.85(x_2 + s_2 + y_2) \quad (6)$$

$$s_i \leq S_i \quad \forall i \in F \quad (7)$$

$$\sum_{j \in A} z_{ij} = x_i \quad \forall i \in F \quad (8)$$

$$\sum_{j \in A} v_{ij} = y_i + s_i \quad \forall i \in F \quad (9)$$

$$w_j = \sum_{i \in F} z_{ij} \quad \forall j \in A \quad (10)$$

$$x_i, s_i, y_i, z_{ij}, v_{ij} \in \mathbb{R}_+ \quad \forall i \in F, j \in A. \quad (11)$$

Uma formulação em AMPL, com um exemplo de um conjunto de dados é

```

set F := 1..2;
set A := 1..2;
set P := 1..2;

param L { F };
param R { P, A };
param S { F };
param p { F };
param h { A };
param c { F, A };

var x { F } >= 0;
var y { F } >= 0;
var s { F } >= 0;
var z { F, A } >= 0;
var v { F, A } >= 0;
var w { A } >= 0;

minimize CustoTotal:
    sum { i in F } p[i]*(x[i]+s[i]+y[i])
    + sum { j in A } w[j]*h[j]
    + sum { i in F, j in A } c[i,j]*(z[i,j]+v[i,j]);

subject to limiteProducao1 { i in F }:
    x[i]+s[i] <= L[i];
subject to limiteProducao2 { i in F }:
    y[i] <= L[i];
subject to quantidadeMinimal { j in A }:
```

```

    sum { i in F } z[i,j] >= R[1,j];
subject to quantidadeMinima2 { j in A }:
    sum { i in F } (z[i,j]+v[i,j]) >= R[2,j];
subject to producaoMinima:
    x[1]+s[1]+y[1] >= 0.85*(x[2]+s[2]+y[2]);
subject to limiteArmazenFabrica { i in F}:
    s[i] <= S[i];
subject to fluxo1 { i in F }:
    sum { j in A } z[i,j] = x[i];
subject to fluxo2 { i in F }:
    sum { j in A } v[i,j] = y[i]+s[i];
subject to fluxo3 { j in A }:
    sum { i in F } z[i,j] = w[j];

data;

param L := 1 100, 2 100;
param R := 1 1 20, 1 2 20, 2 1 80, 2 2 80;
param S := 1 10, 2 10;
param p := 1 6, 2 8;
param h := 1 2, 2 1;
param c := 1 1 10, 1 2 8, 2 1 7, 2 2 11;

end;
```
